Environmental Science

HP COMPUTER CURRICULUM

Air Pollution

TEACHERS ADVISOR

HEWLETT PACKARD

HEWLETT hp PACKARD

Hewlett-Packard
Computer Curriculum Series

environmental science
TEACHER'S ADVISOR

air pollution

by Herbert D. Peckham
Gavilan College

edited by Christine Doerr
Hewlett-Packard

Hewlett-Packard Company
11000 Wolfe Road
Cupertino, California 95014

HP 5951-4468

January 1973

The Hewlett-Packard Computer Curriculum Series represents the published results of a Curriculum Development project sponsored by the Data Systems Division of Hewlett-Packard Company. This project is under the directorship of Jean H. Danver.

This material is designed to be used with any Hewlett-Packard system with the BASIC programming language such as the 9830A Educational BASIC, and the 2000 and 3000 series systems.

The Hewlett-Packard Company makes no warranty, expressed or implied, and assumes no responsibility in connection with the operation of the program material attached hereto.

Printed in the U.S.A.

TABLE OF CONTENTS

INTRODUCTION

This Environmental Science Set of the Hewlett-Packard Computer Curriculum Series consists of a set of a Student Lab Book and a corresponding Teacher's Advisor. It was designed to help meet the need for computer-oriented problems in environmental science, providing students an opportunity to use the power and versatility of the computer to attack some relevant contemporary problems.

The materials are designed for flexible use as desired by the individual instructor. The material in this unit is intended to confront the student with the central issues in one of the most pressing problems in the United States today: namely, air pollution. The automobile, industry, and overpopulation are discussed as the causes of this ecological crisis. The unit emphasizes the "model" approach while studying the role of each of these factors in the overall problem. The material can be used to supplement and enrich your curriculum in any fashion you choose. For example, it can be assigned for individual study or as a class group project.

The Student Book provides text material, reference data, and programming exercises for the students. The Teacher's Advisor contains a brief discussion of the important elements of each exercise, as well as sample programs where appropriate.

Each exercise requires that the students make reasonable assumptions about a given pollution problem and then generate a computer model which describes the situation. These exercises are geared for general education use for secondary or beginning college students, but they do vary in degree of difficulty. Some students will need a good deal of assistance in order to complete the more difficult problems. Though the models are difference equation approximations to systems of coupled first order differential equations, the equations are actually very simple to solve numerically. Only introductory elements of algebra are required to complete the unit and *no* students should be excluded on the basis of mathematical prerequisites. You will, however, need to see that adequate computer time is available to your students. A good planning figure is 30 minutes per students per week. Some students will always want to use a great deal more than the allotted time, and they should be encouraged if at all possible.

For best results, you should study all the solutions until you are certain you have a complete grasp of the general methods. You will note that there are several distinct models presented. You will undoubtedly think of different models. Encourage your students to do the same. There are no *approved* solutions. The more initiative the students show in their modelling procedures, the more chance they have of arriving at a meaningful solution.

PROBLEM MODELLING

The model approach used in this unit is extremely powerful and can be applied to a wide range of topics. The student who becomes intrigued with this idea may want to push on. An excellent reference is "The Man-Made World," Engineering Concepts Curriculum Project, Polytechnic Institute of Brooklyn, McGraw Hill Book Company.

COMPONENTS OF AIR POLLUTION

The standard nomenclature for air pollution was introduced in this unit, even though most of your students have probably not completed a chemistry class. If any of your students would like to do some research in general pollution problems, you might refer them to *Ecology, Pollution, Environment*, by Turk, Turk, and Wittes, published by W. B. Saunders Co., 1972.

THE AUTOMOBILE AND AIR POLLUTION

EXERCISE 1 – Estimating Number of Cars

The key to this exercise is that one square mile contains 640 acres. The student should estimate the amount of land per dwelling. Remember that the area in streets and alleys must be included. Suppose we use the figure of 1/2 acre per dwelling. This means that there should be about 1200 dwellings in the square mile. If we can assume that there are 1.5 cars per dwelling, then there should be about 1800 cars in the area. It doesn't seem unreasonable to assume that about 5% or 90 cars are running on the average at any time. A good round number to use would be 100 cars running at any given time.

This number is susceptible to measurement if students want to count houses and measure traffic flow. A wide range of latitude in assumptions still produces something of the order of 100 cars running per square mile on the average. Students will more than likely come up with slightly different numbers, but small differences aren't important.

EXERCISE 2 – A Simple Model

The algorithm is

$$P_{new} = P_{old} + (R)(N).$$

We begin with $P = 0$. From Table 2, a car traveling at 40 mph produces 76.6 ft^3 of CO per mile. Thus $R = (76.6\ ft^3/mile)(40\ miles/hr)$ or 3064 $ft^3/hour$. From Exercise 1, $N = 100$. The students should plot the results to see that the algorithm is a linear relationship.

Program Listing

```
100  REM EXERCISE-2
110  PRINT
120  PRINT "TIME","CARBON MONOXIDE"
130  PRINT "(HOURS)","(CUBIC FEET)"
140  PRINT
150  LET P=0
160  LET R=3064
170  LET N=100
180  FOR T=0 TO 24
190  PRINT T,P
200  LET P=P+R*N
210  NEXT T
999  END
```

Sample Run

```
TIME            CARBON MONOXIDE
(HOURS)         (CUBIC FEET)

 0               0
 1               306400.
 2               612800.
 3               919200.
 4               1.22560E+06
 5               1.53200E+06
 6               1.83840E+06
 7               2.14480E+06
 8               2.45120E+06
 9               2.75760E+06
 10              3.06400E+06
 11              3.37040E+06
 12              3.67680E+06
 13              3.98320E+06
 14              4.28960E+06
 15              4.59600E+06
 16              4.90240E+06
 17              5.20880E+06
 18              5.51520E+06
 19              5.82160E+06
 20              6.12800E+06
 21              6.43440E+06
 22              6.74080E+06
 23              7.04720E+06
 24              7.35360E+06
```

EXERCISE 3 – Computing Concentrations

The volume of air in millions of cubic feet is given by

$$V = (5280)(5280)(500)/10^6.$$

If there are P cubic feet of CO then the concentration of carbon monoxide in parts per million is given by

$$C = P/V$$

Program Listing

```
100  REM EXERCISE-3
110  PRINT
120  PRINT "TIME","CO CONCENTRATION"
130  PRINT "(HOURS)","(PPM)"
140  PRINT
150  LET P=0
160  LET R=3064
170  LET N=100
180  LET V=5280*5280*500/1.00000E+06
190  FOR T=0 TO 24
200  PRINT T,P/V
210  LET P=P+R*N
220  NEXT T
999  END
```

Sample Run

TIME (HOURS)	CO CONCENTRATION (PPM)
0	0
1	21.9812
2	43.9624
3	65.9435
4	87.9247
5	109.906
6	131.887
7	153.868
8	175.849
9	197.831
10	219.812
11	241.793
12	263.774
13	285.755
14	307.736
15	329.718
16	351.699
17	373.68
18	395.661
19	417.642
20	439.624
21	461.605
22	483.586
23	505.567
24	527.548

EXERCISE 4 – Lethal Concentration

If the CO concentration is about 500 ppm at the end of one day in our model, it should reach 1000 ppm at the end of two days. Obviously, our model isn't accurate, since people would be dying from carbon monoxide on a mass basis in residential areas.

We have assumed that the block of air is 500 feet high. Why not 1000 or 2000? Also, no air moves in or out in the model, but there must be some ventilation. As was pointed out in the introduction to the components of air pollution, something unknown is happening to the carbon monoxide. About the only conclusion the student could reach at this point is that the simple model can't be correct!

EXERCISE 5 – A Garage Problem

Here the situation is quite different than that in Exercise 4, and much more deadly! A production rate of CO of 3064 cubic feet per hour for a car operating at 40 mph is equivalent to about 1 cubic foot per second. A reasonable assumption about garage size would be 20 by 15 by 10 feet or 3 thousand cubic feet. Thus, if a car engine is operating in our closed

garage with an equivalent rate of 40 mph, the CO concentration will pass 1000 ppm in 3 seconds!

There most certainly is a hazard in this situation. If the garage is tightly closed with no ventilation, you can be certain that there is a lethal concentration of CO if an automobile engine is running. If you remain in the garage long enough for the carbon monoxide to be assimilated into the hemoglobin of your blood, that's it! Depending on the concentration of CO, this could happen in a very few minutes. A concentration of 1000 ppm produces unconsciousness in one hour and death in four hours. However, recall that in the garage model, we passed 1000 ppm in 3 seconds!

This example points out the clear and present danger of any open combustion taking place in a closed space. Even a charcoal hibachi burning on a dining room table can be a threat. Certainly, automobile mufflers that leak CO or defective household heating systems that leak CO into the hot air system are extremely dangerous.

EXERCISE 6 – Intersecting Freeways

We must add the average number of cars expected to be on the freeways inside our one square mile residential district. Assume four lanes in each direction constitutes a major freeway. Does this seem reasonable? If, on the average, the cars are spaced a quarter mile apart (certainly much closer during rush hours, and certainly further apart at night), we should expect to find 8 cars in each quarter mile section of the freeway, or 32 cars per freeway per mile. Since we had two intersecting freeways, we must now account for 64 cars operating in the square mile of residential district in addition to the 100 in the original model.

In Exercise 4 we decided that the model was faulty and couldn't be giving valid results. However, there certainly is some air pollution problem and whatever that is, it is increased by 64% when two freeways intersect in the residential district!

EXERCISE 7 – A Tunnel Problem

Let's make the worst case assumptions. Assume there are two lanes of traffic moving at 40 mph. At that rate it will take 90 seconds to pass through the tunnel. If we assume a car is 15 feet long, and there are two car lengths to the car in front, then two cars (one in each lane) are in an air space about 50 feet long, 25 feet wide (?) and possibly 15 feet high (?). This block of air has a volume of about 20 thousand cubic feet. Your students will likely make different assumptions but the end results should agree reasonably well.

It makes no difference if we follow the block of air and two cars together through the tunnel, or if we examine a fixed block of air and consider the cars moving through it. The result is the same in either case. The two cars in the block of air are producing CO at the rate of 2 cubic feet per second (see Exercise 5). Therefore, in 10 seconds the CO concentration should pass the lethal level of 1000 ppm, which means that you would be exposed to higher than lethal levels for about 80 seconds if the ventilation system on the tunnel went out just as your car entered.

Of course, the real hazard takes place if the ventilation system is lost and the traffic stalls! It should be obvious that in such a situation, the motors of the stalled cars should be shut off immediately. This exercise should stimulate a good discussion in your class. It certainly points out how you can begin to get at seemingly hopeless problems with even a little information.

EXERCISE 8 – A New Model

We must make some assumptions to estimate the number of cars in the block of air at any given time. If the cars are 15 feet long, separated by two car lengths, then about every 50 feet on the freeway we will pick up eight cars (one in each of the eight lanes). Two thousand feet of freeway would hold (2000/50) (8) or about 300 cars. However, we have two freeways so there should be about 600 cars in the system. From Table 2, R = 3455 grams per hour.

Program Listing

```
100  REM EXERCISE 8
110  PRINT
120  PRINT "TIME","POLLUTANT CONCENTRATION"
130  PRINT "(HOURS)","(MILLIGRAMS PER CUBIC FOOT)"
140  PRINT
150  LET P=0
160  LET N=600
170  LET R1=3455
180  LET W=5
190  LET R2=1.00000E-02
200  LET T=0
210  LET V=2000*2000*500
220  LET C=1000*P/V
230  PRINT T,C
240  LET P=P+R1*N-W*P/50-R2*P
250  LET T=T+1
260  GOTO 220
999  END
```

Sample Run

TIME (HOURS)	POLLUTANT CONCENTRATION (MILLIGRAMS PER CUBIC FOOT)
0	0
1	1.0365
2	1.95898
3	2.78
4	3.5107
5	4.16102
6	4.73981
7	5.25493
8	5.71339
9	6.12141
10	6.48456
11	6.80776
12	7.0954
13	7.35141
14	7.57925
15	7.78204
16	7.96251
17	8.12314
18	8.26609
19	8.39332
20	8.50656
21	8.60733
22	8.69703
23	8.77685
24	8.8479
25	8.91113
26	8.96741
27	9.01749
28	9.06207
29	9.10174
30	9.13705
31	9.16848
32	9.19644
33	9.22133
34	9.24349
35	9.26321
36	9.28075
37	9.29637
38	9.31027
39	9.32264
40	9.33365

EXERCISE 9 – Equilibrium Concentration

$$P_{eq} = P_{eq} + R_1 N - \frac{W}{50} P_{eq} - R_2 P_{eq}$$

$$0 \quad = R_1 N - P_{eq}\left(\frac{W}{50} + R_2\right)$$

$$\text{or} \quad P_{eq} = (R_1 N)/(W/50 + R_2)$$

$$\text{and} \quad C \quad = 1000 P_{eq}/V$$

Inserting the numerical values of the problem, C is determined to be 9.36, which agrees quite closely with the computer result of Exercise 8.

EXERCISE 10 – Turning the Wind Off

All we have to do to modify the program from Exercise 8 is to put in a counting loop. The new program is shown below. The corresponding printout follows.

Program Listing

```
100  REM EXERCISE  10
110  PRINT
120  PRINT "TIME","POLLUTANT CONCENTRATION"
130  PRINT "(HOURS)","(MILLIGRAMS PER CUBIC FOOT)"
140  PRINT
150  LET P=0
160  LET N=600
170  LET R1=3455
180  LET W=5
190  LET R2=1.00000E-02
200  LET T=0
210  LET V=2000*2000*500
220  LET C=1000*P/V
230  PRINT T,C
240  LET P=P+R1*N-W*P/50-R2*P
250  LET T=T+1
260  IF T <= 10 THEN 220
270  LET W=0
280  IF T >= 21 THEN 999
290  GOTO 220
999  END
```

Sample Run

```
TIME          POLLUTANT CONCENTRATION
(HOURS)       (MILLIGRAMS PER CUBIC FOOT)

 0             0
 1             1.0365
 2             1.95898
 3             2.78
 4             3.5107
 5             4.16102
 6             4.73981
 7             5.25493
 8             5.71339
 9             6.12141
 10            6.48456
 11            6.80776
 12            7.77618
 13            8.73492
 14            9.68407
 15            10.6237
 16            11.554
 17            12.4749
 18            13.3867
 19            14.2893
 20            15.1829
```

Note that at 11 hours the concentrations build up at a steeper rate because of the dissipation mechanism (wind) being turned off. If the program is continued, a new equilibrium value will be attained.

EXERCISE 11 – Closing Down Freeways

We can easily modify the program of Exercise 10 by setting N = 0 instead of W = 0 in line 270. Note in the printout that as soon as the source of pollutants is gone (N = 0), the dissipation mechanisms start cutting down the pollutant concentrations.

Program Listing

```
100  REM EXERCISE 11
110  PRINT
120  PRINT "TIME","POLLUTANT CONCENTRATION"
130  PRINT "(HOURS)","(MILLIGRAMS PER CUBIC FOOT)"
140  PRINT
150  LET P=0
160  LET N=600
170  LET R1=3455
180  LET W=5
190  LET R2=1.00000E-02
200  LET T=0
210  LET V=2000*2000*500
220  LET C=1000*P/V
230  PRINT T,C
240  LET P=P+R1*N-W*P/50-R2*P
250  LET T=T+1
260  IF T <= 10 THEN 220
270  LET N=0
280  IF T >= 21 THEN 999
290  GOTO 220
999  END
```

Sample Run

```
TIME           POLLUTANT CONCENTRATION
(HOURS)        (MILLIGRAMS PER CUBIC FOOT)

 0              0
 1              1.0365
 2              1.95898
 3              2.78
 4              3.5107
 5              4.16102
 6              4.73981
 7              5.25493
 8              5.71339
 9              6.12141
 10             6.48456
 11             6.80776
 12             6.0589
 13             5.39242
 14             4.79926
 15             4.27134
 16             3.80149
 17             3.38333
 18             3.01116
 19             2.67993
 20             2.38514
```

EXERCISE 12 – A Time Dependent Model

The model itself is easy enough to program. The profile factors must be introduced as subscripted variables. This data is read in from DATA statements in the program. The first number in each DATA statement is X, the second Y, and the third Z.

The students can use any profiles they desire. The set of profiles in the program show low traffic during hours of darkness building up to peaks in the morning and afternoon. The wind comes up during the late afternoon and dies back down in the evening. The pollutant disassociation factor is controlled by presence or absence of sunlight.

```
100  REM EXERCISE 12
110  PRINT
120  PRINT "TIME","POLLUTANT CONCENTRATION"
130  PRINT "(HOURS)","(MILLIGRAMS PER CUBIC FOOT)"
140  PRINT
150  DIM X[24],Y[24],Z[24]
160  FOR I=1 TO 24
170  READ X[I],Y[I],Z[I]
180  NEXT I
190  LET P=0
200  LET N=600
210  LET R1=3455
220  LET W=15
230  LET R2=.05
240  LET V=2000*2000*500
250  PRINT 0,0
260  FOR T=1 TO 24
270  LET P=P+R1*X[T]*N-Y[T]*W*P/50-Z[T]*R2*P
280  LET C=1000*P/V
290  PRINT T,C
300  NEXT T
801  DATA .01,0,1
802  DATA .01,0,1
803  DATA .01,0,1
804  DATA .01,0,1
805  DATA .05,0,1
806  DATA .1,0,.7
807  DATA .3,0,.5
808  DATA .75,0,.3
809  DATA .6,0,.1
810  DATA .5,0,.1
811  DATA .4,0,.1
812  DATA .4,0,.1
813  DATA .4,.1,.1
814  DATA .4,.2,.1
815  DATA .5,.5,.1
816  DATA .6,.9,.1
817  DATA .75,1,.3
818  DATA .3,1,.5
819  DATA .2,.8,.7
820  DATA .1,.4,1
821  DATA .1,.1,1
822  DATA .05,0,1
823  DATA .01,0,1
824  DATA .01,0,1
999  END
```

TIME (HOURS)	POLLUTANT CONCENTRATION (MILLIGRAMS PER CUBIC FOOT)
0	0
1	.010365
2	2.02117E-02
3	2.95662E-02
4	3.84528E-02
5	8.83552E-02
6	.188913
7	.49514
8	1.26509
9	1.88066
10	2.38951
11	2.79216
12	3.1928
13	3.49565
14	3.68303
15	3.63041
16	3.25395
17	3.00633
18	2.34022
19	1.90396
20	1.68394
21	1.65287
22	1.62205
23	1.55132
24	1.48412

The results clearly show the effects of the various parts of the model. Up to about 5 a.m., nothing happens. Then the morning rush hour and the rising sun kicks everything off. Note that we would normally expect to get a peak in pollution corresponding to the evening rush hour, but the late afternoon wind prevents this. This model is quite powerful and you can spend a great deal of time experimenting with your class. It should be particularly useful to make plots varying one parameter at a time to get families of concentrations versus time.

EXERCISE 13 – Political Questions

The program from Exercise 14 can be used in this exercise by merely changing the information in the DATA statements. Have the students think a good deal to come up with profile factors that are characteristic of the area in which you live. Make reasonable assumptions about numbers of cars. It would be informative to put a few additional statements into the problem to permit the program to run for several days of simulation. Thus, if all the pollutants are not gone at midnight, the starting value for the next day is not zero, and pollutant "stacking" from day to day can be observed. Also, students can simulate particular days which could pose a problem, for example, a day without wind.

The question of reducing concentrations is designed to force students to consider the implications of different courses of action. It is fine to decree that the number of cars will be restricted, or the freeways will be closed during pollution alerts, etc., but what are the political and economic effects of such decisions? Encourage students to look for realistic decisions which improve the pollution problem and which could actually be implemented from a political and economic point of view.

INDUSTRY AND AIR POLLUTION

EXERCISE 14 – A Zoning Request

It certainly would be foolish to approve a zoning request like this. Notice that the heavy industry section is surrounded by residential areas and parks. We don't know what type of industries are in the heavy and light industry zonings. We don't know what constitutes a commercial zone. There simply isn't enough information to make any decision at all.

EXERCISE 15 – Zoning Modification

The situation is the same as in Exercise 14 except that we are free to arrange things to do as well as possible. The point is that this implies some criteria as to what constitutes "best." The students should decide upon reasonable criteria, e.g., keeping the heavy industry as far away from residential and park areas as possible, and then use the criteria in locating the different functions in the city.

EXERCISE 16 – More Information

Students should use their imagination and what has been learned to this point in this exercise. What type industries are in the zonings? What businesses are in the commercial zone? Is the park and recreation zone polluting? What type of pollutants will be expected to come from the residential zones? Are there any future plans for annexing to the city that could have an effect?

EXERCISE 17 – A Zoning Model

First, we must assume typical numbers and types of activities in each of the nine segments. Knowing this, we can compute the total production of pollutants in each square mile (tons per hour) from the data in Table 3. In this solution we will assume:

LI	*= Jet Aircraft maintenance + meat packing + 5 electronic plants*
	= 3 tons per day or 0.13 tons per hour.
R	*= 100 cars = 760 lbs/hr or 0.38 tons per hour.*
P	*= Non-polluting or 0 tons per hour*
HI	*= Oil refinery + electricity generating plan + railroad shop*
	= 64 tons per day or 2.66 tons per hour.
C	*= 200 cars or 0.76 tons per hour.*

Your students will surely come up with different assumptions than those above. Just check to see that their assumptions are reasonable.

Program Listing

```
10  REM EXERCISE 17
20  DIM R[3,3],P[3,3],N[3,3]
30  FOR I=1 TO 3
40  FOR J=1 TO 3
50  READ R[I,J]
60  NEXT J
70  NEXT I
80  FOR I=1 TO 3
90  FOR J=1 TO 3
100  LET P[I,J]=0
110  NEXT J
120  NEXT I
130  FOR I=1 TO 3
140  FOR J=1 TO 3
150  LET N[I,J]=0
160  NEXT J
170  NEXT I
180  FOR T=1 TO 10
190  LET N[1,1]=P[1,1]+R[1,1]-.2*P[1,1]+.1*(P[2,1]+P[1,2])
200  LET N[2,1]=P[2,1]+R[2,1]-.3*P[2,1]+.1*(P[1,1]+P[2,2]+P[3,1])
210  LET N[3,1]=P[3,1]+R[3,1]-.2*P[3,1]+.1*(P[2,1]+P[3,2])
220  LET N[1,2]=P[1,2]+R[1,2]-.3*P[1,2]+.1*(P[1,1]+P[2,2]+P[1,3])
230  LET N[2,2]=P[2,2]+R[2,2]-.4*P[2,2]+.1*(P[2,1]+P[1,2]+P[2,3]+P[3,2])
240  LET N[3,2]=P[3,2]+R[3,2]-.3*P[3,2]+.1*(P[3,1]+P[2,2]+P[3,3])
250  LET N[1,3]=P[1,3]+R[1,3]-.2*P[1,3]+.1*(P[1,2]+P[2,3])
260  LET N[2,3]=P[2,3]+R[2,3]-.3*P[2,3]+.1*(P[2,2]+P[1,3]+P[3,3])
270  LET N[3,3]=P[3,3]+R[3,3]-.2*P[3,3]+.1*(P[3,2]+P[2,3])
280  FOR I=1 TO 3
290  FOR J=1 TO 3
300  LET P[I,J]=N[I,J]
310  NEXT J
320  NEXT I
330  PRINT
340  PRINT "T = ";T
350  PRINT
360  FOR I=1 TO 3
370  FOR J=1 TO 3
380  PRINT P[I,J],
390  NEXT J
400  PRINT
410  PRINT
420  NEXT I
430  NEXT T
440  DATA .13,.38,.76
450  DATA 0,2.66,.38
460  DATA .38,.38,.13
470  END
```

Sample Run

T = 1

.13	.38	.76
0	2.66	.38
.38	.38	.13

T = 2

.272	1.001	1.444
.317	4.37	1.001
.722	.963	.31

T = 3

.4794	1.6893	2.1154
.7583	5.6102	1.6931
1.0856	1.5943	.5744

T = 4

.75828	2.38301	2.79056
1.24833	6.59962	2.39517
1.48374	2.22303	.91826

T = 5

1.09976	3.06295	3.47027
1.75799	7.44473	3.08746
1.91413	2.83628	1.32643

T = 6

1.4919	3.72554	4.15126
2.27646	8.20131	3.76537
2.37073	3.43393	1.78352

```
T =  7

 1.92372          4.37233          4.8301

 2.79991          8.90092          4.42937

 2.84762          4.01931          2.27674

T =  8

 2.3862           5.0061           5.50425

 3.32717          9.56264          5.08133

 3.34002          4.59604          2.79626

T =  9

 2.87229          5.62958          6.17214

 3.8579           10.1986          5.72325

 3.84434          5.16712          3.33475

T =  10

 3.37658          6.24501          6.833

 4.39206          10.817           6.35683

 4.35797          5.73476          3.88683
```

The matrix R has pollutant production rates in each segment corresponding to the activities as outlined above. Matrix P is the old level of pollutants, and matrix N is the new level. Lines 160 through 240 apply the algorithm to each of the nine segments. The balance of the program follows without difficulty.

The sample run shows that the pollutants from the heavy industry segment diffuse out into the other segments as the hours increase. Since there are no losses in the system as a whole, the pollutants will keep on increasing in each segment and will gradually approach the same concentration at the same time in each of the segments. At this point, the city is "full." Of course, losses should be built in to make the model more reasonable. You could easily assume a wind loss adjustment to the model as was done previously. Encourage your better students to experiment with the model.

EXERCISE 18 – Restructure of the City

This is an open-ended exercise. Students must come up with some criteria to use in judging the worth of a given distribution of activities. An example of such a criteria might be to keep the pollutant levels as low as possible for as long as possible in the residential areas. With such an assumption, students can shift the activities around by modifying the DATA statements in the program of Exercise 18, running the program and analyzing the results.

EXERCISE 19 – Model Your Own Town

This is another open-ended exercise. Here your students must apply most of the knowledge they have acquired in preceding exercises. You should make sure that assumptions are reasonable and that the resultant model corresponds fairly well to the real town that is being modelled. It may not be possible to make a rectangular model of your town in one-mile squares, so feel free to modify the basic model.

PEOPLE AND AIR POLLUTION

EXERCISE 20 – People and Pollution

Since we have no dissipation mechanisms, we may as well consider the pollutant concentration per square mile. We will consider the air over each square mile up to an altitude of 500 feet. The program follows without difficulty.

Program Listing

```
100  REM EXERCISE 20
110  LET V=5280*5280*500
120  FOR N=5000 TO 25000 STEP 5000
130  PRINT
140  PRINT "POPULATION = ";N;"DAILY RATES PER SQUARE MILE"
150  PRINT
160  PRINT "PART","ORG","SO","NO","CO"
170  PRINT
180  LET P1=(N/25)*9.00000E-02*454000./V
190  LET O1=(N/25)*4.5/(V/1.00000E+06)
200  LET S1=(N/25)*.7/(V/1.00000E+06)
210  LET N1=(N/25)*2.6/(V/1.00000E+06)
220  LET C1=(N/25)*32.6/(V/1.00000E+06)
230  PRINT P1,O1,S1,N1,C1
240  NEXT N
999  END
```

Sample Run

```
POPULATION =  5000    DAILY RATES PER SQUARE MILE

PART           ORG            SO             NO             CO

 5.86260E-04    6.45661E-02    1.00436E-02    3.73049E-02    .467746

POPULATION =  10000   DAILY RATES PER SQUARE MILE

PART           ORG            SO             NO             CO

 1.17252E-03    .129132        2.00872E-02    7.46097E-02    .935491

POPULATION =  15000   DAILY RATES PER SQUARE MILE

PART           ORG            SO             NO             CO

 1.75878E-03    .193698        3.01309E-02    .111915        1.40324

POPULATION =  20000   DAILY RATES PER SQUARE MILE

PART           ORG            SO             NO             CO

 2.34504E-03    .258264        4.01745E-02    .149219        1.87098

POPULATION =  25000   DAILY RATES PER SQUARE MILE

PART           ORG            SO             NO             CO

 2.93130E-03    .322831        5.02181E-02    .186524        2.33873
```

EXERCISE 21 – Dissipation Mechanisms

Open-ended problem – turn your students loose!

EXERCISE 22 – A Segmented Model

Open-ended. Use some of the approaches developed previously. Encourage imagination and hard work.

EXERCISE 23 – A Real City

The point here is that, as one proceeds inland from the ocean, the air pollution problem worsens and approaches some kind of bad limit as you approach Riverside. If the model is reasonable, it must account for this!

THE FUTURE

EXERCISE 24 – Pollution In Your Home Town

This is a good exercise in civics and air pollution. Students should be very curious and fairly knowledgeable about air pollution at this point.

EXERCISE 25 – Pollution Control Programs

This exercise follows from Exercise 24.

NOTES

NOTES

NOTES